



Optimizarea funcționării unui sistem de lacuri

Neculai Andrei

*Institutul de Cercetări în Informatică,
Centrul de Modelare și Optimizare Avansată
8-10, Bdl. Averescu, București 1, România,
E-mail: nandrei@ici.ro*

Rezumat. Prezentăm optimizarea funcționării bazinelor hidrografice ale unor râuri care alimentează câteva lacuri și rezervoare (bazine). Din rezervoare apa este luată pentru irigații conform unui grafic de debit cunoscut pe cele 12 luni ale unui an. Problema constă în a aloca apa din râuri, lacuri și rezervoare în așa manieră încât utilizatorii să aibă debitul de apă impus corespunzător lunii respective. Se prezintă modelul și expresia lui în GAMS, împreună cu o situație concretă care se referă la 2 lacuri, 5 rezervoare și 2 utilizatori.

1. Introducere

Un bazin hidrografic include mai multe râuri, lacuri, rezervoare (bazine) construite artificial, precum și utilizatori ai apei care în principal o utilizează la irigații. În situații normale problema managementului unui bazin hidrografic se referă la satisfacerea nevoilor de apă a unor utilizatori, de-a lungul unei perioade de timp (de obicei un an).

Pentru a modela un bazin hidrografic acesta se poate reprezenta ca un graf. Nodurile acestui graf sunt anumite elemente ale bazinului, iar laturile sunt canalele de comunicație între noduri. Fiecare nod este caracterizat de o regulă de modificare a debitului de apă care constă din balanța de substanță (apă).

Pentru a preciza lucrurile să considerăm următorul exemplu de bazin hidrografic, prezentat în figura 1.

Evident că se pot prezenta situații mai complexe, cu mai multe elemente și cu mai multe conexiuni între acestea. Totuși, cazul prezentat este suficient de general pentru a ilustra tehnica de modelare și reprezentarea acestuia în limbajul GAMS.

Pentru reprezentarea matematică a unui bazin hidrografic considerăm următoarele mulțimi de noduri:

- nn - noduri ale rezervoarelor (Rez1, Rez2, ..., Rez5),
- ns - noduri ale surselor de apă, ale râurilor (Sursa1, Sursa2),
- nr - noduri ale utilizatorilor (Utilizator1, Utilizator2, Evacuare),
- nrr - noduri ale utilizatorilor (Utilizator1, Utilizator2)
- nl - noduri ale lacurilor (Lac1, Lac2).




Fig. 1. Bazin hidrografic cu 2 râuri, 2 lacuri, 5 rezervoare și 2 utilizatori.

Considerăm că orizontul de optimizare este de un an și că dispunem de date privind debitul de apă din râuri (*sursa*) în lacuri, ca valori medii lunare, ca în tabelul de mai jos:

	Ian	Feb	Mar	Apr	Mai	Iun
Râu1	98	115	244	390	641	754
Râu2	29	49	78	121	198	144
	Iul	Aug	Sep	Oct	Nov	Dec
Râu1	807	512	367	210	181	128
Râu2	105	98	79	72	45	29

Evident că aceste valori medii sunt cunoscute în urma unor măsurători efectuate de-a lungul anilor și care presupune o situație normală a regimului hidrografic.

Definim:

$R(n, t)$ - ieșirea din nodul n la momentul de timp t (luna t) (m^3),

$Q(n, t)$ - intrarea în nodul n la momentul de timp t (luna t) (m^3),

$S(n, t)$ - stocul (volumul) de apă din nodul n la momentul de timp t (luna t) (m^3),

$U(n, t)$ - Cantitatea de apă pentru utilizatorul n la momentul de timp t (luna t) (m^3).

Cu acestea ecuația lacurilor este:

$$S(n, t) = S(n, t - 1) + \sum_{n \in in} Q(n, t) - \sum_{n \in out} R(n, t), \quad n \in nl,$$

unde *in* reprezintă numărul de intrări în lac, iar *out* numărul de ieșiri din lac. Se presupune că se cunoaște stocul inițial de apă din lacuri.

Ecuția rezervoarelor este următoarea:

$$\sum_{n \in out} R(n, t) = \sum_{n \in in} Q(n, t), \quad n \in nn,$$

Nodurile de irigare (utilizator) sunt caracterizate de ecuația:

$$R(n, t) = ret_n \sum_{n \in in} Q(n, t), \quad n \in nrr,$$

unde ret_n este un coeficient de utilizare asociat fiecărui nod de irigare.

Funcția obiectiv se exprimă în funcție de nevoile de apă la nodurile utilizator. Presupunem că de-a lungul celor 12 luni cele două noduri utilizator au nevoie de următoarele debite (*cerere*) de apă:

	Ian	Feb	Mar	Apr	Mai	Iun
U1	0	0	10	64.5	189.8	184.4
U2	0	0	10	13.5	15	22.1
	Iul	Aug	Sep	Oct	Nov	Dec
U1	243.7	200.9	99.5	0	0	0
U2	26	24.9	13	0	0	0

Atunci, funcția obiectiv se scrie sub forma:

$$\min \sum_n \sum_t (U(n, t) - cerere(n, t))^2.$$

2. Modelul în GAMS

Expresia GAMS a modelului bazinului hidrografic din figura 1 este arătată în figura 2.

```

$ontext
Octombrie 6, 2005

                                MODEL DE LACURI

Optimizarea functionarii bazinelor hidrografice ale unor rauri
care contine cateva lacuri.
In acest exemplu am considerat doua surse (izvoare, rauri) care
alimenteaza doua lacuri. Din aceste doua lacuri apa se duce
intr-un lant de 5 rezervoare puse in cascada. Din rezervorul 2
se ia apa pentru utilizatorul 1, iar din rezervorul 4 se ia apa
pentru utilizatorul 2. Utilizatorii pot folosi apa pentru
irigatii, de exemplu. Din ultimul rezervor (al 5-lea) apa este
deversata intr-un evacuator, care fizic poate fi chiar un parau
care apartine bazinului hidrografic considerat. Pentru a avea
control asupra bazinului, evacuatorul il vom considera ca un
utilizator (special).
Debitul de apa (m3 per secunda) care vine din surse (izvoare,

```

rauri) este cunoscut pe fiecare luna, ca valori medii.
 In acelasi timp se cunoaste cererea (nevoia) de apa, (debitul de apa pentru irigatii, m3 per sec) pentru fiecare utilizator, in fiecare luna.

Problema este de a determina fluxurile de apa din acest bazin hidrografic asa fel incat debitul de apa la utilizatori sa fie cel impus. Altfel spus problema este de a minimiza deviatia cantitatii de apa data utilizatorilor fata de cererea (nevoia) de apa impusa.

\$offtext

```
SET n noduri
  / Sursa_1, Sursa_2,
    Rez_1*Rez_5,
    Utilizator_1, Utilizator_2,
    Lac_1, Lac_2,
    Evacuare /;
```

ALIAS (n,n1);

```
SET
nn(n)  Rezervoare      / Rez_1*Rez_5/
ns(n)  Surse           / Sursa_1*Sursa_2 /
nr(n)  Utilizator      / Utilizator_1, Utilizator_2, Evacuare
/
nrr(n)  Irigari        / Utilizator_1, Utilizator_2 /
nl(n)  Lacuri          / Lac_1, Lac_2 /;
```

SET n_from_n(n,n1) Topologia rețelei de Lacuri - Rezervoare -
 * Evacuare. De unde se ia apa. De exemplu: Lacul 1 ia apa din
 * sursa 1, Lacul 2 din sursa 2, etc.

```
/ Lac_1.Sursa_1,
  Lac_2.Sursa_2,
  Rez_1.Lac_1,
  Rez_1.Lac_2,
  Rez_2.Rez_1,
  Utilizator_1.Rez_2,
  Rez_3.Rez_2,
  Rez_3.Utilizator_1,
  Rez_4.Rez_3,
  Utilizator_2.Rez_4,
  Rez_5.Rez_4,
  Rez_5.Utilizator_2,
  Evacuare.Rez_5 /;
```

```
SET n_to_nr(n,n1)
* Topologia rețelei care alimentează utilizatorii.
/ Rez_2.Utilizator_1,
  Rez_4.Utilizator_2,
  Rez_5.Evacuare /;
```

SET t luni /Ian, Feb, Mar, Apr, Mai, Iun, Iul, Aug, Sep, Oct, Nov, Dec /;

```
PARAMETER Ini_S(n) Cantitatea initiala de apa din Lacuri. (m3)
/ Lac_1 1000,
  Lac_2 300 /;
```

```
PARAMETER ret(n) Coeficientii utilizatorilor
/ Utilizator_1 0.5,
  Utilizator_2 0.5,
```

```

Evacuare      0.0 /;

TABLE Sursa(n,t) Debitul de apa (m3 per sec) din surse.
      Ian Feb  Mar  Apr  Mai  Iun  Iul  Aug  Sep  Oct  Nov
Dec
Sursa_1   98 115 244 390 641 754 807 512 367 210 181
128
Sursa_2   29  49  78 121 198 144 105  98  79  72  45
29

TABLE Cerere(n,t) Cererea de apa (m3 per sec) ceruta de
utilizatori.
      Ian Feb  Mar  Apr  Mai  Iun  Iul  Aug
Sep Oct Nov Dec
Utilizator_1      0  0  10  64.5 189.8 184.4 243.7 200.9
99.5  0
0  0
Utilizator_2      0  0  10  13.5 15.0 22.1 26.0 24.9
13.0  0
0  0
Evacuare      500 500 500 100 100 100 100 500
500 500 500 500

POSITIVE VARIABLES
Utilizator(n,t) Cantitatea de apa de la nodul n in perioada t
(m3)
q(n,t)          Intrare in nodul n in perioada t (m3)
r(n,t)          Iesire din nodul n in perioada t (m3)
s(n,t)          Stoc (volumul) de apa din nodul n in perioada t
(m3);

Variable obj;

* Marginile superioare ale Utilizatorilor.
Utilizator.up(n,t) = Cerere(n,t);

* Marginile superioare ale Lacurilor.
s.up('Lac_1',t) = 1000;
s.up('Lac_2',t) = 300;

* Stocul final in Lacuri (din decembrie).
s.lo('Lac_1','Dec') = 1000;
s.lo('Lac_2','Dec') = 300;

EQUATIONS
R_no(n,t)      Noduri simple
R_ns(n,t)      Surse (rauri)
R_nr(n,t)      Utilizatori (de irigare)
R_nl(n,t)      Rezervoare
R_nn(n,t)      Noduri simple
Objective      Functia obiectiv;

* Ecuatiile pentru nodurile simple: iesire = intrare
R_no(n,t)$ (nn(n)).. R(n,t) =e= q(n,t);

* Ecuatiile pentru surse: iesire = Sursa
R_ns(n,t)$ (ns(n)).. R(n,t) =e= Sursa(n,t);

* Ecuatiile Utilizatorilor (de irigare): iesire = coef *
Utilizator

```

```

R_nr(n,t)$nr(n).. R(n,t) =e= ret(n) * Utilizator(n,t);

* Ecuatiile Rezervoarelor: iesire = bilantul de masa (apa)
R_nl(n,t)$nl(n).. S(n,t) =e= Ini_S(n)$ord(t) eq 1
                    + s(n,t-1)$ord(t) gt 1)+Q(n,t)-
R(n,t);

* Ecuatiile nodurilor simple: intrare = suma iesirilor din
rezervoarele
* din amonte minus suma iesirilor din rezervoare
R_nn(n,t).. Q(n,t) =e= sum(nl$(n_from_n(n,nl)), R(nl,t))
                - sum(nl$(n_to_nr(n,nl))
                ,Utilizator(nl,t));
*Functia obiectiv
objective.. obj =e= sum(t,
                    sum(n$nr(n), power((Utilizator(n,t) -
                    Cerere(n,t)),2) ));

MODEL Lacuri /all/;

SOLVE Lacuri USING NLP minimizing obj;

* Afisarea rezultatelor optimizarii modelului de lacuri

file Rezultate /lac1.txt /
put Rezultate;
put "Nod      Utiliz_1  Cerere_1  Utiliz_2  Cerere_2
Evacuare    Cerere"/;
loop((t), put t.tl:6,
        Utilizator.l('Utilizator_1',t):11.2,
Cerere('Utilizator_1',t):11.2,
        Utilizator.l('Utilizator_2',t):11.2,
Cerere('Utilizator_2',t):11.2,
        Utilizator.l('Evacuare',t): 11.2, Cerere('Evacuare',t)
:11.2 /;);

put /"----- Lacuri -----
----- "/;

put "
Lac 2 "/;
put "      intrare      stoc      iesire      intrare
stoc      iesire "/;
put "      Q-1      S-1      R-1      Q-2
S-2      R-2 "/;
loop((t), put t.tl:8,
        Q.L('Lac_1',t):11.2, S.L('Lac_1',t):11.2,
R.L('Lac_1',t):11.2,
        Q.L('Lac_2',t):11.2, S.L('Lac_2',t):11.2, R.L('Lac_2',t):11.2
/;);
* End of model

```

Fig. 2. Modelul în GAMS.

3. Soluția modelului

Rezultatele optimizării cu MINOS sunt date în tabelele de mai jos:

Nod	Utiliz_1	Cerere_1	Utiliz_2	Cerere_2	Evacuare	Cerere
Ian	0.00	0.00	0.00	0.00	500.00	500.00

Feb	0.00	0.00	0.00	0.00	500.00	500.00
Mar	10.00	10.00	10.00	10.00	500.00	500.00
Apr	64.50	64.50	13.50	13.50	100.00	100.00
Mai	189.80	189.80	15.00	15.00	100.00	100.00
Iun	184.40	184.40	22.10	22.10	100.00	100.00
Iul	243.70	243.70	26.00	26.00	100.00	100.00
Aug	199.93	200.90	23.93	24.90	498.07	500.00
Sep	58.00	99.50	0.00	13.00	417.00	500.00
Oct	0.00	0.00	0.00	0.00	282.00	500.00
Nov	0.00	0.00	0.00	0.00	226.00	500.00
Dec	0.00	0.00	0.00	0.00	157.00	500.00

----- Lacuri -----

	Lac 1			Lac 2		
	intrare Q-1	stoc S-1	iesire R-1	intrare Q-2	stoc S-2	iesire R-2
Ian	98.00	524.00	574.00	29.00	0.00	329.00
Feb	115.00	188.00	451.00	49.00	0.00	49.00
Mar	244.00	0.00	432.00	78.00	0.00	78.00
Apr	390.00	0.00	390.00	121.00	0.00	121.00
Mai	641.00	0.00	641.00	198.00	51.00	147.00
Iun	754.00	436.70	317.30	144.00	195.00	0.00
Iul	807.00	1000.00	243.70	105.00	300.00	0.00
Aug	512.00	1000.00	512.00	98.00	300.00	98.00
Sep	367.00	1000.00	367.00	79.00	300.00	79.00
Oct	210.00	1000.00	210.00	72.00	300.00	72.00
Nov	181.00	1000.00	181.00	45.00	300.00	45.00
Dec	128.00	1000.00	128.00	29.00	300.00	29.00

Figurile 3 și 4 ilustrează evoluția variabilelor (intrări, ieșiri, stocuri) asociate lacurilor.




Fig. 3. Evoluția variabilelor asociate Lacului 1.




Fig. 4. Evoluția variabilelor asociate Lacului 2.

Referințe

- Daene McKinney and Andre Savitsky, *Basic optimization models for water and energy management*, The University of Texas at Austin Technical Report, 2003.
- Brooke, A., Kendrick, D., Meeraus, A., Raman, R. and Rosenthal, R.E., *GAMS A user's guide*. GAMS Development Corporation, December 1998.

Octombrie 6, 2005